2.2 Hypothesis about the number of factors and key items

EFA Variations

- Hypothesis about the number of factors:
  - ANALYSIS: TYPE = EFA
  - ESEM (*1)
  - PSEM with GEOMIN priors
  - Second-order exploratory factor analysis (SEFA) using PSEM with GEOMIN priors
  - Bi-factor analysis using ROTATION = BI-GEOMIN and direct second-order exploratory factor analysis (DSEFA) using PSEM
- **Hypothesis about the number of factors and key items**:
  - **ESEM with Target rotation**
  - PSEM with ALF priors for cross loadings
- Comparing EFA methods
- Special models:
  - ESEM with PSEM priors for residual covariances
  - PSEM finding a small number of cross-loadings

Slide 45 returns to the EFA Variations overview where we now turn to analyses where not only the number of factors is specified but also key items. This was referred to as the middle analysis stage on slide 4.

The first topic is ESEM with target rotation.

2.2.1 ESEM with Target rotation

ESEM with Target Rotation

- A stage in between of EFA and CFA: Rotation guided by substantive considerations, not using mechanical EFA rotation
- Same model fit as EFA, just a different rotation
- Choose rotation by giving target loading values (typically zero)
- Target values not fixed as in CFA — zero targets can come out big if misspecified
- Minimum requirement for identification is $m-1$ (m = # factors) zeros in each loading column which gives EFA which together with factor variances fixed at 1 results in the m^2 restrictions $m(m-1)+m$
- Mplus language using ESEM:
  - ```
    f1 BY y1-y10 y1~0 (*1);
    f2 BY y1-y10 y5~0 (*1);
    ```
- Tucker (1944), Browne (1972a, b). Asparouhov & Muthén (2024) suggests that PSEM does better than Target

Slide 46 discusses Target rotation. Target rotation represents an analysis stage that lies in between of EFA and CFA: The rotation is guided by substantive considerations, not using mechanical EFA rotation.

It has the same model fit as EFA, just using a different rotation. You choose a rotation by giving target loading values (typically zero).

Target values are not fixed as in CFA — zero targets can come out big if misspecified.

Minimum requirement for identification is $m-1$ (where m = #factors) zeros in each loading column which gives EFA which together with factor variances fixed at 1 results in the $m(m-1)+m = m^2$ restrictions.

The Mplus language uses ESEM for one factor at a time using the label 1 to denote the same EFA block. A curl $\sim$ is used together with a value to denote a target value for a loading, typically zero.

At the bottom are two early references for this technique and also the 2024 PSEM paper by us. As we will see, there are now better approaches to target rotation.
    ```

ESEM Target Input and Results for the H&S example

```
ANALYSIS:  ESTIMATOR = MLR;
           ROTATION = TARGET;

MODEL:     spatial BY visual-flags general-figurew~0 (*1);
           verbal BY visual-flags~0 general-wordm addition-figurew~0 (*1);
           speed BY visual-wordm~0 addition-straight wordr-figurew~0 (*1);
           memory BY visual-straight~0 wordr-figurew (*1);
           spatial-memory@1;
```

Grant-White has 9 significant cross-loadings, Pasteur has 12

Slide 47 shows the target input for the H&S example with 19 variables. The ANALYSIS uses ROTATION = TARGET. The MODEL command uses the loading pattern of slide 9 to give zero target values to all cross loadings.

This analysis results in 9 significant cross loadings for the Grant- White school and 12 for Pasteur. This is more than the 5 and 8 that we saw with regular EFA.

ESEM Bi-Factor Target Input for the H&S Example

```
ANALYSIS: ESTIMATOR = MLR;
          ROTATION = TARGET(ORTHOGONAL);

MODEL:    spatial BY visual-flags general-sentence~0 wordc-figurew(*1);
          verbal BY visual-paper~0 flags-figurew(*1);
          speed BY visual-straight wordr-figurer~0 object-figurew(*1);
          memory BY visual-wordm addition-counting~0 straight-figurew(*1);
          g BY visual-figurew(*1);
```

- This will be referred to as Model M5 in a later table

Slide 48 shows that Mplus also offers target rotation with bi-factor modeling. In the ANALYSIS command, this uses ROTATION = TARGET(ORTHOGONAL) so that all factors are uncorrelated. The MODEL command adds the general factor called g to the ESEM specification. On slides 56 and 57 this is the specification referred to as M5.

2.2.2 PSEM with ALF priors for cross loadings

EFA Variations

- Hypothesis about the number of factors:
  - ANALYSIS: TYPE = EFA
  - ESEM (*1)
  - PSEM with GEOMIN priors
  - Second-order exploratory factor analysis (SEFA) using PSEM with GEOMIN priors
  - Bi-factor analysis using ROTATION = BI-GEOMIN and direct second-order exploratory factor analysis (DSEFA) using PSEM
- Hypothesis about the number of factors and key items:
  - ESEM with Target rotation
  - **PSEM with ALF priors for cross loadings**
- Comparing EFA methods
- Special models:
  - ESEM with PSEM priors for residual covariances
  - PSEM finding a small number of cross-loadings

Slide 49 shows the EFA variations overview again and we see that we will now turn to using PSEM with ALF priors for cross loadings.

PSEM with ALF Priors for Cross-Loadings for the 4-Factor H&S19 Example

- Specify the CFA factors
- Label the cross loadings
- Give ALF priors to the cross loadings

```
ANALYSIS:        ESTIMATOR = MLR;

MODEL:           spatial BY visual-flags*1
                       general-figurew*0 (a1-a15);
                 verbal BY visual-flags*0 (b1-b4)
                       general-wordm*1
                       addition-figurew*0 (c1-c10);
                 speed BY visual-wordm*0 (d1-d9)
                       addition-straight*1
                       wordr-figurew*0 (e1-e6);
                 memory BY visual-straight*0 (f1-f13)
                       wordr-figurew*1;
                 spatial-memory@1;

MODEL PRIORS:    a1-f13~ALF(0,1.0);
```


Slide 50 shows the input for PSEM with ALF priors for cross loadings. There are 3 parts to this: Specify the CFA factor loadings, label the cross loadings, and give ALF priors for the cross loadings in MODEL PRIORS.

The cross loadings are the same loadings that were given zero targets in the Target rotation approach. With ALF priors, the targets need not be exactly zero but approximately so by varying the variance of the ALF prior. Target rotation would correspond to using a very small ALF prior variance so that the values are essentially held at the prior mean of zero. The ALF prior variances can also be different for different cross loadings.

The intention of the PSEM ALF cross-loadings analysis is that its log likelihood value should match that of the 4-factor EFA so that PSEM ALF essentially provides a different rotation. To accomplish this, a variance of 1 is often a suitable choice for the ALF prior variance but this should be checked.

For factor indicators that have widely varying variances, standardization of the indicators using the DEFINE command is recommended for the ALF priors to work well.

Comparing EFA, Target, and PSEM ALF for the 4-Factor H&S19 Example

- Log likelihood is the same for the 3 approaches when those runs standardize the variables, but different rotations are used
- Number of significant cross loadings for Grant-White/Pasteur:
  - EFA (GEOMIN): 5/8
  - Target: 9/12
  - PSEM ALF (v=1): 5/7

Slide 51 gives a comparison of EFA, Target rotation, and PSEM ALF cross loadings for the H&S example with 19 variables. The model fit is the same - the log likelihoods agree when the variables are standardized before analysis.

But, the 3 methods use different rotations producing different numbers of significant cross loadings for Grant-White/Pasteur: 5/8, 9/12, 5/7. The PSEM ALF cross loading approach appears to give the simplest solution.

Target Simulations in Asparouhov-Muthén (2024)

- Section 4.4 comparison of PSEM and ESEM alternatives (N = 500)
- This web talk adds PSEM Geomin and ESEM Geomin results to Table 4 (BSEM dropped)

$$\Lambda = \begin{array}{c} \quad F1 \quad F2 \\ \begin{bmatrix} 1 & 0 \\ 1 & 0 \\ 1 & 0 \\ 0 & 1 \\ 0 & 1 \\ 0.3 & 1 \end{bmatrix} \end{array}, \qquad \Psi = \begin{bmatrix} 1 & \\ 0.5 & 1 \end{bmatrix} \tag{1}$$

- Critique of target rotation for the case of incorrect target

Slide 52 describes a simulation study in our 2024 PSEM paper. The simulations support that using PSEM ALF priors for cross loadings is preferred over Target rotation.

The model used for the simulations is shown in equation (1). The simple loading pattern is broken by one cross loading that appears for the last variable and the first factor. The premise is that the Target rotation incorrectly specified a zero target for this loading.

Methods Compared

- PSEM ALF: Cross-loadings given ALF(0,1) priors as 3 slides earlier
- PSEM LASSO: Cross-loadings given LASSO(0,1) priors
- PSEM Normal: Cross-loadings given N(0,1) priors
- PSEM Geomin: All loadings given GEOMIN(2,1,0.0001) priors
- ESEM Geomin (similar to PSEM Geomin):
  - ANALYSIS: ROTATION = GEOMIN;
  - MODEL: f1-f2 BY y1-y6 (*1);
- ESEM Target:
  - ANALYSIS: ROTATION = TARGET;
    MODEL: f1 BY y1-y3 y4-y6~ 0 (*1);
    f2 BY y1-y3~0 y4-y6 (*1);

Slide 53 lists the different approaches that are studied. The preferred PSEM ALF approach is at the top. The target rotation approach is listed last showing the incorrect zero target for the y6 loading of the f1 factor.

Table 4 Extended: Abs Bias(Coverage)

Par.	True Value	PSEM ALF	PSEM LASSO	PSEM Normal	PSEM Geomin	ESEM Geomin	ESEM Target
λ_{11}	1	.00(.93)	.00(.93)	.01(.93)	.00(.94)	.00(.93)	.00(.93)
λ_{12}	0	.00(1.0)	.00(1.0)	.01(.94)	.00(1.0)	.00(.99)	.01(.95)
λ_{61}	.3	.02(.99)	.03(.92)	.09(.27)	.02(.97)	.02(.99)	.09(.27)
λ_{62}	1	.00(.95)	.00(.95)	.02(.91)	.00(.95)	.00(.95)	.02(.91)
ψ_{12}	.25	.00(.96)	.02(.94)	.06(.71)	.00(.95)	.01(.96)	.07(.63)

- Conclusion:
  - PSEM ALF, PSEM GEOMIN, and ESEM GEOMIN perform best
  - PSEM Normal and ESEM Target worst (see λ_{61})
- Target problems and iterative target discussed in Moore, Reise, Depaoli, & Haviland (2015)

Slide 54 shows the table of results. This is Table 4 of Section 4.4 of the 2024 PSEM paper. The critical loading λ_{61} is particularly sensitive to the method used. The target approach (ESEM Target) gets a large bias and poor coverage for this loading.

The overall conclusion is that PSEM ALF, PSEM GEOMIN, and ESEM GEOMIN perform best while PSEM Normal and ESEM Target perform the worst.

Problems with target rotation has been studied in the literature before, with suggestions to use an iterative target approach where target loadings that obtain large estimates are dropped as targets in a second target analysis. An overview of this topic is given in the Moore et al. (2015) paper given in the reference section.